

Úloha č. 3 z Programování pro fyziky

Následující program v Pascalu

```
program Uloha3;

procedure vypisObvod4( xa,ya, xc,yc : real; n:integer );
var x1,x2,y1,y2:real;
begin
  Writeln( xa,' ', ya,' ', n);    // vrchol A
  Writeln( xc,' ', ya,' ', n);    // vrchol B
  Writeln( xc,' ', yc,' ', n);    // vrchol C
  Writeln( xa,' ', yc,' ', n);    // vrchol D
  Writeln( xa,' ', ya,' ', n);    // vrchol A
  Writeln;

  x1 := (2*xa+xc)/3;
  x2 := (xa+2*xc)/3;
  y1 := (2*ya+yc)/3;
  y2 := (ya+2*yc)/3;

  if n>0 then begin
    vypisObvod4( xa,ya, x1,y1, n-1 );
    vypisObvod4( x1,ya, x2,y1, n-1 );
    vypisObvod4( x2,ya, xc,y1, n-1 );
    vypisObvod4( xa,y1, x1,y2, n-1 );
    vypisObvod4( x2,y1, xc,y2, n-1 );
    vypisObvod4( xa,y2, x1,yc, n-1 );
    vypisObvod4( x1,y2, x2,yc, n-1 );
    vypisObvod4( x2,y2, xc,yc, n-1 );
  end;
end;



const n = 5;

begin
  vypisObvod4( 0,0, 1,1, n);
end.
```

vypíše data, která po přeměrování výstupu programu do souboru `ctverec.txt` mohou být vykreslena příkazy `gnuplotu`

```
set size ratio -1
unset border
unset tics
plot 'ctverec.txt' with lines linecolor black
```

Tak získáme obrázek vlevo:



Upravte program tak, aby vypisoval data, jejichž vykreslením vznikne obrázek vpravo. Na něm je rovnostranný trojúhelník rozdělen na čtyři spojením středů jednotlivých stran, přičemž tři z trojúhelníků jsou pak rozdělovány dále.

Až budete spokojeni s výsledným obrázkem, vykreslíte data místo na obrazovku do souboru pomocí příkazů

```
set term pdf
set output "uloha3.pdf"
replot
unset term
```


Provedením výše uvedené posloupnosti příkazů vznikne soubor `uloha3.pdf`.

Odevzdání řešení: Na webové stránce, jejíž adresa je uvedena v SISu v modulu *studijní mezivýsledky*, je třeba do 14. listopadu odevzdat

- Zdrojový kód vašeho programu (v Pascalu, C++, ...) a
- obrázek (ve formátu pdf).

Následuje několik rad, které by mohly urychlit váš postup:

- Místo pěti příkazů `writeln` u čtvece, budou potřena pro trojúhelník jen čtyři.
- Střed strany je aritmetický průměr kartézských souřadnic jejích konců.
- Ačkoli je ze souřadnic vrcholů A a B rovnostranného trojúhelníku možno dopočíst souřadnice vrcholu C, bude snazší proceduře `VypisObvod3` předat rovnou šest parametrů `xa,ya, xb,yb, xc,yc` a nakonec ještě `n`.
- Největší rovnostranný trojúhelník může mít např. vrcholy o souřadnicích $[0, 0]$, $[2, 0]$, $[1, \sqrt{3}]$.
- Pro hodnoty parametru $n = 1, 2, 3, 4$ byste měli získat následující obrázky:



- Neváhejte navštívit stránku https://en.wikipedia.org/wiki/Sierpi%C5%84ski_triangle